类 分区主页 : 窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,宽度=-1,高度=-1,图片资源="../../素材/Image_5809178934246521.jpg",拉伸方式=6]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=相对布局1,宽度=-1,高度=-1,纵向布局=真]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,@上外边距DP=80]])
	变量 相对布局2 : 相对布局
	@布局配置([[父布局=相对布局2,纵向布局=真,@位于布局中间=真]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,内容="仙武传",字体大小=45]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局2,内容="版本V1.0.1",对齐方式=17]])
	变量 文本框2 : 文本框
	@布局配置([[父布局=线性布局1,宽度=-1,@上外边距DP=150]])
	变量 相对布局3 : 相对布局
	@布局配置([[父布局=相对布局3,@位于布局中间=真,ID=1]])
	变量 线性布局3 : 线性布局
	@布局配置([[父布局=线性布局3,宽度=-2,高度=-2,卡片阴影=0,圆角=20]])
	变量 卡片布局1 : 卡片布局
	@布局配置([[父布局=卡片布局1,宽度=-1,高度=-1]])
	变量 线性布局4 : 线性布局
	@布局配置([[父布局=线性布局4,宽度DP=10,高度DP=10,卡片阴影=0,@左外边距DP=10,@上外边距DP=4,圆角=20,@下外边距DP=4]])
	变量 卡片布局2 : 卡片布局
	@布局配置([[父布局=线性布局4,内容="仙武1区",字体大小=12,@上外边距DP=0,@左外边距DP=5,字体颜色=-9065681,@下外边距DP=3]])
	变量 文本框3 : 文本框
	@布局配置([[父布局=线性布局4,内容="切换",@左外边距DP=5,@上外边距DP=0,字体大小=12,字体颜色=-4255437,@右外边距DP=10,@下外边距DP=3]])
	变量 文本框4 : 文本框
	@布局配置([[父布局=线性布局1,宽度=-1,纵向布局=真]])
	变量 线性布局5 : 线性布局
	@布局配置([[父布局=线性布局5,宽度DP=110,高度DP=35,@布局对齐方式=17,@上外边距DP=15,卡片背景颜色=-13487566]])
	变量 卡片布局3 : 卡片布局
	@布局配置([[父布局=卡片布局3,宽度=-1,高度=-1,对齐方式=17,内容="开始游戏"]])
	变量 文本框5 : 文本框
	@布局配置([[父布局=线性布局5,宽度DP=110,高度DP=35,@布局对齐方式=17,@上外边距DP=10,卡片背景颜色=-13487566]])
	变量 卡片布局4 : 卡片布局
	@布局配置([[父布局=卡片布局4,宽度=-1,高度=-1,对齐方式=17,内容="账号管理"]])
	变量 文本框6 : 文本框
	@布局配置([[父布局=线性布局5,宽度DP=110,高度DP=35,@布局对齐方式=17,@上外边距DP=10,卡片背景颜色=-13487566]])
	变量 卡片布局5 : 卡片布局
	@布局配置([[父布局=卡片布局5,宽度=-1,高度=-1,对齐方式=17,内容="删除记录"]])
	变量 文本框7 : 文本框
	@布局配置([[父布局=线性布局5,宽度DP=110,高度DP=35,@布局对齐方式=17,@上外边距DP=10,卡片背景颜色=-13487566]])
	变量 卡片布局6 : 卡片布局
	@布局配置([[父布局=卡片布局6,宽度=-1,高度=-1,对齐方式=17,内容="退出游戏",字体颜色=-457981]])
	变量 文本框8 : 文本框
	@布局配置([[父布局=线性布局1,宽度=-1,高度=-1]])
	变量 相对布局4 : 相对布局
	@布局配置([[父布局=相对布局4,宽度=-1,对齐方式=17,内容="健康游戏警告\n抵制不良游戏，拒绝盗版游戏。\n注意自我保护，谨慎上当受骗。\n适当游戏益脑，沉迷游戏伤身。\n合理安排时间，享受健康生活。\n",@位于布局底部=真]])
	变量 文本框9 : 文本框

	事件 分区主页:创建完毕()
		订阅事件()
		本对象.显示标题栏 =假
		本对象.沉浸模式 = 真
		本对象.显示到异形屏区域()
		本对象.状态栏字体黑色 = 假
		隐藏底部导航栏(本对象)
		文本框1.字体颜色 = 白色
		文本框2.字体颜色 = 白色
		文本框5.字体颜色 = 白色
		文本框6.字体颜色 = 白色
		文本框7.字体颜色 = 白色
		文本框9.字体颜色 = 白色
		卡片布局1.卡片背景颜色 = -1610612736
		卡片布局2.卡片背景颜色 = -15166089
		共享数据.初始化(本对象,"保存")
		如果 读入文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/分区")=="" 则
			文本框3.内容 = "未知分区"
		否则
			文本框3.内容 =读入文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/分区")
		结束 如果

	结束 事件
	//进入游戏
	事件 卡片布局3:被单击()
		变量 获取分区:文本=读入文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/分区")

		//弹出提示(读入文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/"+获取分区+"/"+"昵称"))
		如果 AES解密(共享数据.取文本("token"),keys)=="" 则
			弹出提示("请先登录账号")
			返回
		结束 如果
		如果 读入文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/分区")=="" 则
			弹出提示("请先选择分区")
			返回
		结束 如果

		如果 读入文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/"+获取分区+"/"+"昵称")=="" 则
			切换窗口(创建用户昵称)
			播放默认切换动画(淡入淡出)
			关闭窗口()
		否则
			开始游戏(本对象)
			变量 游戏加载:DX加载
			游戏加载.内容("载入游戏中...")
			游戏加载.显示()
			提交到新线程运行()
			延时(500)
			切换窗口(游戏主页)
			播放默认切换动画(淡入淡出)
			关闭窗口()
			结束提交到新线程()
		结束 如果
	结束 事件

	//登录账号管理
	事件 卡片布局4:被单击()
		切换窗口(账号管理)
		播放默认切换动画(淡入淡出)
		关闭窗口()
	结束 事件
	//删除用户数据
	事件 卡片布局5:被单击()
		变量 提示:对话框
		提示.标题 = "警告"
		提示.信息 = "确认删除用户数据？"
		提示.按钮1 = "确定"
		提示.显示()
		事件 提示:按钮1被单击()

			如果 共享数据.取文本("账号")=="" 则
				弹出提示("请先登录账号")
			否则
				删除文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/"+读入文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/分区"))
				写出文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/分区","")
				文本框3.内容 = "未知分区"
				弹出提示("删除成功")
			结束 如果
		结束 事件

	结束 事件
	//退出app
	事件 卡片布局6:被单击()
		关闭程序()
	结束 事件
	//切换分区列表
	事件 文本框4:被单击()
		切换窗口(全部分区)
		播放默认切换动画(淡入淡出)
		关闭窗口()
	结束 事件
	变量 再按一次退出:逻辑型=假
	变量 重置再按一次退出:时钟
	@虚拟方法
	方法 返回键被按下()
		如果 再按一次退出==假 则
			弹出提示("再按一次退出")
			再按一次退出=真
			重置再按一次退出.时钟周期 = 3000
		否则
			返回桌面()
			关闭程序()
		结束 如果
	结束 方法
	事件 重置再按一次退出:周期事件()
		重置再按一次退出.时钟周期 = 0
		再按一次退出=假
	结束 事件
结束 类
类 全部分区:窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1,纵向布局=真,背景颜色=-13882324]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,@上外边距DP=35]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,宽度DP=30,高度DP=30,图片资源="../../素材/jiantouzuo.png",@左外边距DP=12]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=相对布局1,内容="全部分区",@位于布局中间=真,字体大小=18]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局1,列数=3,高度=-1]])
	变量 自定义宫格列表框1 : 自定义宫格列表框

	变量 分区:文本集合
	变量 状态:文本集合
	事件 全部分区:创建完毕()
		订阅事件()
		本对象.显示标题栏 =假
		本对象.沉浸模式 = 真
		本对象.显示到异形屏区域()
		本对象.状态栏字体黑色 = 假
		隐藏底部导航栏(本对象)
		文本框1.字体颜色 = 白色
		共享数据.初始化(本对象,"保存")
		获取全部分区()
	结束 事件
	事件 自定义宫格列表框1:加载布局(索引 : 整数,项目布局 : 组件容器) : 组件容器
		变量 布局:分区布局
		如果 项目布局 == 空
			布局 = 创建 分区布局
		否则
			布局 = 项目布局
		结束 如果
		布局.文本框1.内容 = 分区[索引]
		事件 布局.取根布局():被单击()
			如果 状态[索引] ? "正常" 则
				弹出提示(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/分区"+"/"+分区[索引])
				//	创建目录(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/"+分区[索引])
				写出文本文件(取数据目录路径()+"/date/"+共享数据.取文本("账号")+"/分区",分区[索引])
				//	切换窗口(分区主页)
				//	播放默认切换动画(淡入淡出)
				//	关闭窗口()
			否则
				弹出提示(状态[索引])
			结束 如果
		结束 事件
		返回 布局
	结束 事件

	@虚拟方法
	方法 返回键被按下()
		切换窗口(分区主页)
		播放默认切换动画(淡入淡出)
		关闭窗口()
	结束 方法
	事件 图片框1:被单击()
		切换窗口(分区主页)
		播放默认切换动画(淡入淡出)
		关闭窗口()
	结束 事件
	方法 获取全部分区()
		变量 获取分区:Oks异步
		获取分区.请求链接(服务器+"cfq.php")
		获取分区.回调主线程(真)
		获取分区.POST()
		事件 获取分区:请求成功(结果 : Oks结果)
			变量 测试 : JSON数组 = 结果.到文本()
			循环(索引, 0, 测试.长度)
				变量 测试_项目 : JSON对象 = 测试[索引]
				变量 name : 文本 = 测试_项目["name"]
				变量 status : 文本 = 测试_项目["status"]
				分区.添加成员(name)
				状态.添加成员(status)
			结束 循环
			自定义宫格列表框1.更新项目(分区.长度)
		结束 事件

	结束 方法

结束 类
类 分区布局:组件容器

	@布局配置([[根布局=真,宽度=-1,高度=-1]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度DP=110,高度DP=40,圆角=20,卡片阴影=1,卡片背景颜色=5066061,背景颜色=-11711155,@左外边距DP=8,@上外边距DP=15,@右外边距DP=8]])
	变量 卡片布局1 : 卡片布局
	@布局配置([[父布局=卡片布局1,宽度=-1,高度=-1,对齐方式=17,字体颜色=-16220331]])
	变量 文本框1 : 文本框

结束 类


类 账号管理:窗口

	@布局配置([[根布局=真,宽度=-1,高度=-1,纵向布局=真,背景颜色=-13882324]])
	变量 线性布局1 : 线性布局
	@布局配置([[父布局=线性布局1,宽度=-1,@上外边距DP=35]])
	变量 相对布局1 : 相对布局
	@布局配置([[父布局=相对布局1,宽度DP=30,高度DP=30,图片资源="../../素材/jiantouzuo.png",@左外边距DP=12]])
	变量 图片框1 : 图片框
	@布局配置([[父布局=相对布局1,内容="登录账号",@位于布局中间=真,字体大小=18]])
	变量 文本框1 : 文本框
	@布局配置([[父布局=线性布局1,内容="若账号不存在时将自动注册，请勿把账号泄露给他人，以免造成不可挽回的损失！",@左外边距DP=12,@上外边距DP=15,@右外边距DP=10]])
	变量 文本框2 : 文本框
	@布局配置([[父布局=线性布局1,宽度=-1,纵向布局=真,@上外边距DP=30,@右外边距DP=12,@左外边距DP=12]])
	变量 线性布局2 : 线性布局
	@布局配置([[父布局=线性布局2,内容="账号:6-15位数字，推荐qq号",@下外边距DP=5]])
	变量 文本框3 : 文本框
	@布局配置([[父布局=线性布局2,左内边距DP=10,提示文本="输入账号",输入方式=2]])
	变量 编辑框1 : 编辑框
	@布局配置([[父布局=线性布局2,内容="密码:6至8位数字或英文",@上外边距DP=8,@下外边距DP=5]])
	变量 文本框4 : 文本框
	@布局配置([[父布局=线性布局2,左内边距DP=10,提示文本="输入密码",输入方式=129]])
	变量 编辑框2 : 编辑框
	@布局配置([[父布局=线性布局1,内容="注册/登录账号代表您已经仔细阅读并同意《用户协议及隐私政策》",字体大小=10,@上外边距DP=40,@左外边距DP=15,@右外边距DP=12,@下外边距DP=5]])
	变量 文本框5 : 文本框
	@布局配置([[父布局=线性布局1,@左外边距DP=12,@右外边距DP=12,内容="登录"]])
	变量 按钮1 : 按钮
	事件 账号管理:创建完毕()
		订阅事件()
		本对象.显示标题栏 =假
		本对象.沉浸模式 = 真
		本对象.显示到异形屏区域()
		本对象.状态栏字体黑色 = 假
		隐藏底部导航栏(本对象)
		文本框1.字体颜色 = 白色
		文本框2.字体颜色 = 白色
		文本框3.字体颜色 = 白色
		文本框4.字体颜色 = 白色
		文本框5.字体颜色 = 白色
		置圆角背景(编辑框1,黑色,15,15,15,15)
		编辑框1.字体颜色 = 白色
		编辑框1.提示文本颜色 = 白色
		置圆角背景(编辑框2,黑色,15,15,15,15)
		编辑框2.字体颜色 = 白色
		编辑框2.提示文本颜色 = 白色
		按钮1.背景颜色 = -16220331
		按钮1.字体颜色 = 白色
		编辑框1.内容 = 共享数据.取文本("账号")
		编辑框2.内容 = AES解密(共享数据.取文本("密码"),keys)
	结束 事件
	事件 按钮1:被单击()
		如果 编辑框1.内容.为空()或 编辑框2.内容.为空() 则
			弹出提示("账号或密码不能为空")
		否则
			登录()
		结束 如果
	结束 事件
	方法 登录()
		变量 登录:Oks异步
		登录.请求链接(服务器+"login.php")
		登录.添请求参数("user",编辑框1.内容)
		登录.添请求参数("pass",编辑框2.内容)
		登录.缓存报文(真)
		登录.回调主线程(真)
		登录.POST()
		事件 登录:请求成功(结果 : Oks结果)
			变量 结果1:文本=结果.到文本()
			如果 结果1 ? "该用户不存在" 则
				//不存在注册
				注册()
			否则
				如果 结果1 ? "密码错误" 则
					弹出提示(结果1)
				否则
					变量 登录成功:JSON对象=结果.到文本()
					变量 status : 文本 = 登录成功["status"]
					变量 token : 文本 = 登录成功["token"]
					弹出提示(status)
					共享数据.置文本("token",AES加密(token,keys))
					共享数据.置文本("账号",编辑框1.内容)
					共享数据.置文本("密码",AES加密(编辑框2.内容,keys))
					切换窗口(分区主页)
					播放默认切换动画(淡入淡出)
					关闭窗口()
				结束 如果
			结束 如果
		结束 事件

	结束 方法

	方法 注册()
		变量 注册账号:Oks异步
		注册账号.请求链接(服务器+"reg.php")
		注册账号.添请求参数("user",编辑框1.内容)
		注册账号.添请求参数("pass",编辑框2.内容)
		注册账号.POST()
		事件 注册账号:请求成功(结果 : Oks结果)
			如果 结果.到文本()=="注册成功" 则
				登录()
			否则
				弹出提示(结果.到文本())
			结束 如果
		结束 事件
	结束 方法
	@虚拟方法
	方法 返回键被按下()
		切换窗口(分区主页)
		播放默认切换动画(淡入淡出)
		关闭窗口()
	结束 方法
	事件 图片框1:被单击()
		切换窗口(分区主页)
		播放默认切换动画(淡入淡出)
		关闭窗口()
	结束 事件

结束 类